

# **ECE 351**

## Verilog and FPGA Design

---

**Week 4\_1:      Questions about Homework #1?**  
**User-defined types (wrap-up)**  
**RTL expression operators**

Roy Kravitz  
Electrical and Computer Engineering Department  
Maseeh College of Engineering and Computer Science



---

## Questions about Homework #1?

Write-up: [..\..\assignments\hw1\ece351s21\\_hw1\\_release\\_R1\\_0\ece351\\_hw1.pdf](..\..\assignments\hw1\ece351s21_hw1_release_R1_0\ece351_hw1.pdf)

Question 1: [..\..\assignments\hw1\ece351s21\\_hw1\\_release\\_R1\\_0\ece351sp1\\_hw1a\\_r1.txt](..\..\assignments\hw1\ece351s21_hw1_release_R1_0\ece351sp1_hw1a_r1.txt)

Stimulus: [..\..\assignments\hw1\ece351s21\\_hw1\\_release\\_R1\\_0\hdl\stimulus.sv](..\..\assignments\hw1\ece351s21_hw1_release_R1_0\hdl\stimulus.sv)

Testbench: [..\..\assignments\hw1\ece351s21\\_hw1\\_release\\_R1\\_0\hdl\tb\\_CLA4Bit.sv](..\..\assignments\hw1\ece351s21_hw1_release_R1_0\hdl\tb_CLA4Bit.sv)

## Review: Enumerated type values

- SystemVerilog allows you to declare a type with an explicit list of valid values
- SystemVerilog by default represents values for enumerated types as `int` with first label represented by value of 0, second label with value of 1, etc.
  - User can override (e.g. to map values to specific hardware – one-hot, Gray code, etc)
  - Not required to specify all values
    - Unspecified values continue numbering from previous value

```
enum {ONE = 1, FIVE = 5, TEN = 10 } state;
enum {A=1, B, C, X=24, Y, Z} list1;
enum {A=1, B, C, D=3} list2; // ERROR
```

- SystemVerilog permits an explicit base type (with size)

```
// enumerated type with a 1-bit wide,
// 2-state base type
enum bit {TRUE, FALSE} Boolean;

// enumerated type with a 2-bit wide,
// 4-state base type
enum logic [1:0] {WAITE, LOAD, READY} state;

enum logic [2:0] {WAITE = 3'b001,
LOAD = 3'b010,
READY = 3'b100} state;
```

## Using typedefs w/ enum(s)

- Creating a typedef allows creation of multiple variables of same enumerated type in different places

```
typedef enum {WAITE, LOAD, READY} states_t;  
  
states_t state, next_state;
```

- Enumerated types are semi-strongly and can only be assigned:
  - A label from its enumerated type list
  - Another enumerated type of the same type (declared with same **typedef** definition)
  - A value cast to the **typedef** type of the enumerated type

## Using typedefs w/ enum(s) (cont'd)

---

```
typedef enum {WAITE, LOAD, READY} states_t;
states_t state, next_state;
int foo;

state = next_state;    // Legal: both variables of same enumerated type
foo = state + 1;        // Legal: foo is of type int, underlying enum is int
state = foo + 1;        // Illegal: foo is type int, foo+1 is int not enum type
state = state + 1;      // Illegal: (state + 1) type is int not enum type
state++;               // Illegal: (state + 1) type is int not enum type
next_state += state;   // Illegal: (state + 1) type is int not enum type
```

## System tasks & methods for enumerated types<sup>7</sup>

SystemVerilog provides several special system functions called methods to iterate through the values in an enumerated type list

```
<enum_var>.first      // Return value of first member in enumerated
                      // list of var
<enum_var>.last       // Return value of last member in enumerated
                      // list of var
<enum_var>.next(<N>)  // Return value of next member in enumerated
                      // list. If N provided return Nth next member
<enum_var>.prev(<N>)  // Return value of previous member in enumerated
                      // list.If N provided return Nth previous member
<enum_var>.num        // Return the number of labels in the enumerated
                      // list of var
<enum_var>.name       // Return string representation of label for
                      // value
```

## Review: Parameterized Types

---

```
module adder #(parameter type dtype = logic [0:0]) // default is 1-bit size
(
    input dtype a, b,
    output dtype sum
);

assign sum = a + b;

endmodule

module top (
    input logic [15:0] a, b,
    input logic [31:0] c, d,
    output logic [15:0] r1,
    output logic [31:0] r2
);

adder #(.dtype(logic [15:0])) i1 (a, b, r1); // 16 bit adder
adder #(.dtype(logic signed [31:0])) i2 (c, d, r2); // 32-bit signed adder

endmodule
```

---

## User defined types – Struct(ures) and Unions

Source material drawn from:

- Mark F. and Roy K. ECE 571 lecture slides
- *RTL Modeling with SystemVerilog* by Stuart Sutherland
- *Logic Design and Verification Using SystemVerilog* by Donald Thomas



## Struct(ure)s

---

```
typedef enum {Request, Response, Broadcast} PacketType;
```

```
typedef struct {  
    int ID;  
    PacketType Type;  
    int CheckSum;  
    byte Data[1024];  
} Packet_t;
```

```
Packet_t SamplePacket;
```

```
SamplePacket.ID = 0;  
SamplePacket.Type = Request;
```

```
.  
.   
.
```

## Packed struct(s)

- Packed structs give you more control over how bits are laid out in memory

```
typedef struct {bit [7:0] r, g, b;} pixel_s;  
pixel_s my_pixel;
```


Consumes 3 words

```
typedef struct packed {bit [7:0] r, g, b;} pixel_p_s;  
pixel_p_s my_pixel;
```

Consumes 3 bytes


## Packed struct(s) (cont'd)

```
reg [24:0] Entry;
`define Valid 24
`define Tag 23:16
`define Addr 15:0
iTag = Entry[`Tag];
iAddr = Entry[`Addr];
iValid = Entry[`Valid]
```



unpacked

```
struct packed {
    bit Valid;
    byte Tag;
    bit [15:0] Addr;
} Entry;
iTag = Entry.Tag;
iAddr = Entry.Addr;
iValid = Entry.Valid
```



packed

## Initializing struct(s)

---

```
initial begin
    typedef struct {int a;
                    byte b;
                    shortint c;
                    int d;} my_struct_s;
    my_struct_s st = '{32'haaaa_aaaad,
                      8'hbb,
                      16'hcccc,
                      32'hdddd_dddd};

    $display("str = %x %x %x %x ", st.a, st.b, st.c, st.d);
end
```

# Unions

---

- Stores several types (mutually exclusive fields) in the same bits

```
typedef union { int i; real f; } num_u;  
num_u un;  
un.f = 0.0; // set value in floating point format
```

## Summary

---

- ❑ typedef, enum, struct, union were added to SystemVerilog (i.e. did not exist in Verilog 2001)
- ❑ Modeled after same constructs in C
- ❑ User-defined types (typedef)
  - Allow users to define new types built from predefined types or other user-defined types
  - Can be used as module ports and passed in or out of tasks and functions
- ❑ Enumerated types (enum)
  - Allow the declaration of variables with a define set of values (represented by abstract labels instead of hardware-centric logic values)
  - Allow modeling at a more abstract level than Verilog 2001
  - Values of labels can be specified (ex: one-hot encoding)
  - Default type is int...should always size and type your enums
  - Are strongly typed compared to other SystemVerilog constructs
  - Have a set of functions to get at elements of the enum, etc.

## Summary (cont'd)

---

### □ Structures (struct)

- Make it possible to bundle several variables together and work w/ the complete bundle while still being able to work w/ the individual variables
- Structs can be copied, assigned list of values, passed through module ports, and passed in and out of functions and tasks
- Can be packed or unpacked...better to use packed when you can

### □ Unions (union)

- Give high-level coding style for modeling shared resources in a design
- Ex: a register that can store different types of data at different times
- Can be packed or unpacked. Use packed if based on packed struct(s)

## RTL Expression Operators

Source material drawn from:

- Roy's ECE 351 and ECE 571 lecture material
- *RTL Modeling with SystemVerilog* by Stuart Sutherland
- *Logic Design and Verification Using SystemVerilog* by Donald Thomas



## Expressions operators and operands

- RTL and behavioral modeling describe a design in expressions instead of primitive gates
- Expressions are constructs that combine operators and operands to produce a result
  - ex: `a ^ b; addr1[20:17] + addr2[20:17]; in1 | in2;`
- Operands can be any of the SystemVerilog data types
  - constants, integers, real numbers, wires, logic, vectors or parts of vectors, function calls
- Operators act on operands to produce desired results
  - `count + 1; a && b; !P1; reg[3:0] >> 2;`
- 2-state and 4-state expressions
  - When any of the operand is a 4-state expression the result of the operation will be a 4-state expression
  - All operands must be 2-state expressions for an expression to have a 2-state result
  - Recommended Guideline: Only use 4-state types for RTL modeling

## X-optimization and X-pessimism

- Most SystemVerilog operators are X-optimistic...will produce a known result even if there are X or Z values in the operands

```
logic [3:0] a, b, c, result;
assign a = 4'b01zx;      // some bits are X or Z
assign b = 4'b0000;      // all bits are zero
assign c = 4'b1111;      // all bits are one

assign result = a & b;    // result is 4'b0000 because & models a
                          // AND gate and 0 ANDed w/ any value is 0
assign result = a & c;    // result is 4'b01xx because 1 ANDed w/
                          // any value could result in either a 0
                          // or a 1
```

## X-optimization and X-pessimism (cont'd)

- Arithmetic and relational operators are X-pessimistic...will produce an X result if any operand has any bit w/ an X or Z value

```
logic [3:0] a, b, c, result;  
assign a = 4'b000x;      // some bits are X or Z  
assign b = 4'b0101;      // all bits are zero  
assign c = 4'b1111;      // value of 5 in decimal  
  
assign result = a + b;    // result is 4'bxxxx because + adds  
                        // numbers and 4'b000x is not a number
```

## Vector sizes

- Self-determined operands – operators treat each operand independently even if the vectors are different sizes
  - ex: && does a logical which tests to see if both operands are true – each operand can be evaluate to be true or false independent of the vector size of other operand
- Context-determined operands – operators need to expand the operands to be the same vector size before the operation is performed
  - Operation will left-extend the shortest operand to be the same vector size as the largest operand
  - ex: & does a bitwise AND and returns a Boolean (1'b1 or 1'b0) result of each bit
  - Rules:
    - If leftmost bit is 0 or 1 and operand is unsigned than zero-extend
    - If leftmost bit is Z than operand is Z-extended
    - If leftmost bit is X than operand is X-extended

## Vector sizes (cont'd)

---

- Signed and unsigned expressions
  - If all operands are signed than a signed operation is performed
  - If any of the operands are unsigned than an unsigned operation is performed
- Integer (vector) and real (floating-point)
  - Operations can be performed on mix of integer and real expressions
  - Rule: If any of the operands is a real expression than the other operand is converted to a real-expression and a floating-point operation is performed

RTL Synthesis compilers typically do not support real (floating-point) expressions

# SystemVerilog operators

| Category    | Examples  | Bit Length                         |
|-------------|---|------------------------------------|
| Bitwise     | $\sim A$ ,<br>$A \& B$ , $A   B$ , $A \wedge B$ , $A \wedge \sim B$                     | $L(A)$<br>$\text{MAX}(L(A), L(B))$ |
| Logical     | $!A$ , $A \&\& B$ , $A    B$  | 1 bit                              |
| Reduction   | $\&A$ , $\sim\&A$ , $ A$ , $\sim  A$ , $\wedge \sim A$ , $\sim \wedge A$                | 1 bit                              |
| Relational  | $A == B$ , $A != B$ , $A > B$ , $A < B$<br>$A >= B$ , $A <= B$<br>$A === B$ , $A !== B$ | 1 bit                              |
| Arithmetic  | $A + B$ , $A - B$ , $A * B$ , $A / B$<br>$A \% B$                                       | $\text{MAX}(L(A), L(B))$           |
| Shift       | $A \ll B$ , $A \gg B$   | $L(A)$                             |
| Concatenate | $\{A, \dots, B\}$   | $L(A) + \dots + L(B)$              |
| Replication | $\{B\{A\}\}$  | $B * L(A)$                         |
| Condition   | $A ? B : C$   | $\text{MAX}(L(B), L(C))$           |

## Operator precedence

### □ Operators default precedence

+ , - , ! , ~ (unary)

+ , - (Binary)

<< , >>

< , > , <= , >=

== , !=

&

^ , ^~ or ~^


|

&&

||

?: (ternary)

Highest Priority



Lowest Priority

### □ Parenthesis ( ) can be used to override defaults

## Concatenation and replication

---

| Operator             | Example Usage           | Description  |
|----------------------|-------------------------|--|
| <code>{ }</code>     | <code>{m, n}</code>     | Join <i>m</i> and <i>n</i> together as a vector  |
| <code>{r{ } }</code> | <code>{r{m, n} }</code> | Join <i>m</i> and <i>n</i> together, and replicate <i>r</i> times; <i>r</i> must be a literal integer value. it cannot be a parameter. |



## Concatenation and replication (cont'd)

- ❑ Converse of bit/part select:
  - `logic[9:0] CBus;`
  - `assign Cbus[7:0] = {Apart, 3'b000, Abit};`
- ❑ Can be used to replicate:
  - `Logic [9:0] CBus;`
  - `assign Cbus = { 6{Abit}, 3'b000};`
- ❑ The operands may be scalar nets or registers, vector nets or registers, bit-select, part-select or sized constants
  - Must be sized because the size of each operand must be known for computation of the size of the result
- ❑ Are synthesizable
- ❑ Do not directly add hardware but simply combine, append, or break apart vectors

## Concatenation and replication (cont'd)

```
//
// 8-bit status register that stores multiple input values //
// -----
// | int_en | unused | unused | zero | carry | neg | parity |
// -----
// NOTE: not-used bits are set to a constant 1
//
module status_reg (
    input logic clk,                // register clock
    input logic rstN,               // active-low reset
    input logic int_en,             // 1-bit interrupt enable
    input logic zero,               // 1-bit result = 0 flag
    input logic carry,              // 1-bit result overflow flag
    input logic neg,                // 1-bit negative result flag
    input logic [1:0] parity,       // 2-bit parity bits
    output logic [7:0] status       // 8-bit status register output
);

always_ff @(posedge clk or negedge rstN) // async reset
    if (!rstN)                          // active-low reset
        status <= {1'b0,2'b11,5'b0};    // reset
    else
        status <= {int_en,2'b11,zero,carry,neg,parity}; // load
endmodule: status_reg
```



## Conditional operator ?:

### □ Implies a multiplexer

```
module cond (sel, a, b, y);
    input sel, a, b;
    output y;
    assign y = sel ? a : b;
endmodule
```

### □ Can be nested:

```
input s1, s0;
input i3, i2, i1, i0;

//4-to-1 multiplexer
assign y = s1?(s0?i3:i2)
           :(s0?i1:i0);
```

### □ Can model a tri-state driver:

```
module tri_buf (en, a, y);
    input en, a;
    output y;
    assign y = en ? a : 1'bz;
endmodule
```

## Bitwise operators

29

| Operator                           | Operation       | Examples<br><code>ain = 3'b101, bin = 3'b110, cin = 3'b01x</code>                                     |
|------------------------------------|-----------------|---|
| <code>~</code>                     | invert each bit | <code>~ain</code> is <code>3'b010</code>  |
| <code>&amp;</code>                 | and each bit    | <code>ain &amp; bin</code> is <code>3'b100</code> , <code>bin &amp; cin</code> is <code>3'b010</code> |
| <code> </code>                     | or each bit     | <code>ain   bin</code> is <code>3'b111</code>   |
| <code>^</code>                     | xor each bit    | <code>ain ^ bin</code> is <code>3'b011</code>   |
| <code>~^</code> or <code>^~</code> | xnor each bit   | <code>ain ^~ bin</code> = <code>3'b100</code>   |

- ☐ Operates on each bit of the operand
- ☐ Result is the size of the largest operand
- ☐ Left-extended if sizes are different
- ☐ Bitwise operators are X-optimistic

ECE 351 Verilog and FPGA Design



## Bitwise operators (cont'd)

**Table 5-5:** Bitwise AND truth table

| & | 0 | 1 | x | z |
|---|---|---|---|---|
| 0 | 0 | 0 | 0 | 0 |
| 1 | 0 | 1 | x | x |
| x | 0 | x | x | x |
| z | 0 | x | x | x |

**Table 5-6:** Bitwise OR truth table

|   | 0 | 1 | x | z |
|---|---|---|---|---|
| 0 | 0 | 1 | x | x |
| 1 | 1 | 1 | 1 | 1 |
| x | x | 1 | x | x |
| z | x | 1 | x | x |

**Table 5-7:** Bitwise XOR truth table

| ^ | 0 | 1 | x | z |
|---|---|---|---|---|
| 0 | 0 | 1 | x | x |
| 1 | 1 | 0 | x | x |
| x | x | x | x | x |
| z | x | x | x | x |

**Table 5-8:** Bitwise exclusive NOR truth table

| $\sim \sim$ | 0 | 1 | x | z |
|-------------|---|---|---|---|
| 0           | 1 | 0 | x | x |
| 1           | 0 | 1 | x | x |
| x           | x | x | x | x |
| z           | x | x | x | x |


## Reduction operators

**Table 5-9:** Reduction operators for RTL modeling

| Operator               | Example Usage              | Description                 |
|------------------------|----------------------------|-----------------------------|
| <b>&amp;</b>           | <b>&amp; m</b>             | AND all bits of m           |
| <b>~&amp;</b>          | <b>~&amp; m</b>            | NAND all bits of m          |
| <b> </b>               | <b>  m</b>                 | OR all bits of m            |
| <b>~ </b>              | <b>~  m</b>                | NOR all bits of m           |
| <b>^</b>               | <b>^ m</b>                 | Exclusive-OR all bits of m  |
| <b>^^</b><br><b>^~</b> | <b>^^ m</b><br><b>^~ m</b> | Exclusive-NOR all bits of m |

- ❑ Models gates yielding a single output bit
- ❑ Reduction operators are X-Optimistic

```
logic[3:0] a;
Logic b;
assign b = &a;
```



## Reduction operators (cont'd)

```
// User-defined type definitions
package definitions_pkg;
  typedef struct {
    logic [3:0] data;
    logic      parity_bit;
  } data_t;
endpackage: definitions_pkg

// Parity checker using even parity (the combined data value
// plus parity bit should have an even number of bits set to 1
module parity_checker
  import definitions_pkg::*;
  (input data_t data_in, // 5-bit structure input
   input clk,           // clock input
   input rstN,          // active-low asynchronous reset
   output logic error    // set if parity error detected
  );

  always_ff @(posedge clk or negedge rstN) // async reset
    if (!rstN) error <= 0;                // active-low reset
    else      error <= ^{data_in.parity_bit, data_in.data};
    // reduction-XOR returns 1 if an odd number of bits are
    // set in the combined data and parity_bit
endmodule: parity_checker
```

## Next Time

---

- ☐ Topics:
  - RTL expression operators (wrap-up)
  - RTL programming statements
- ☐ You should:
  - Review Sutherland Ch 5
  - Read Sutherland Ch 6
- ☐ Homework, projects and quizzes
  - Homework #1 has been posted. Should be submitted to D2L by 10:00 PM on Mon, 26-Apr
    - ☐ No late assignments accepted after Noon on Thu, 29-Apr
  - **In-class exercise pushed to Tue, 27-Apr**